

SỬ DỤNG MAPLE TRONG HÌNH HỌC SƠ CẤP

Phan Đức Châu

Nói chung MAPLE không thể dùng làm công cụ để chứng minh trong Hình học, bởi vì khi gán các tọa độ tổng quát cho các điểm, ta có quá nhiều tham số cần đặt điều kiện. MAPLE chỉ dùng để chứng minh trong các trường hợp cụ thể. Tuy nhiên, trong một số bài toán ta cũng có thể sử dụng MAPLE để chứng minh tổng quát được.

I. Sử dụng MAPLE chứng minh định lý trong trường hợp cụ thể

1. Ví dụ 1

Định lý (Đường tròn Euler)

Trong một tam giác bất kỳ, chín điểm sau nằm trên một đường tròn: 3 chân đường cao, 3 trung điểm các cạnh, 3 trung điểm đoạn thẳng nối đỉnh với trực tâm.

Đường tròn này có tên Đường tròn Euler.


Chọn một tam giác ABC có tọa độ cụ thể, chẳng hạn $A(2,4)$, $B(0,0)$, $C(7,0)$. Vẽ đường tròn (c) đi qua chân 3 đường cao. Sau đó chứng minh 6 điểm còn lại nằm trên (c). Định lý được chứng minh. Các tọa độ đỉnh được chọn tùy ý, sao cho 3 điểm đó không thẳng hàng.

Bổ sung thêm bằng hình vẽ minh họa.

Gõ các lệnh sau (Nhấn Enter sau các lệnh):

```
> restart : with(geometry) :  
triangle(T, [point(A, 2, 4), point(B, 0, 0), point(C, 7, 0)]) :  
midpoint(M1, B, C) : midpoint(M2, C, A) : midpoint(M3, A, B) :  
orthocenter(H, T) :  
altitude(AH1, A, T, H1) : altitude(BH2, B, T, H2) : altitude(CH3, C,  
T, H3) : circle(c, [H1, H2, H3]) :  
midpoint(N1, A, H) : midpoint(N2, B, H) : midpoint(N3, C, H) :  
draw([T(color = blue), c, H1, H2, H3, H], axes = none, printtext  
= true, title = 'Duong tron di qua chan ba duong cao');  
"Kiem tra cac diem con lai nam tren duong tron nay";
```

Duong tron di qua chan ba duong cao



"Kiểm tra các điểm còn lại nằm trên đường tròn này"

```

if IsOnCircle(M1, c) = true then print('M1 ∈ c') end if;
                                     M1 ∈ c

if IsOnCircle(M2, c) = true then print('M2 ∈ c') end if;
                                     M2 ∈ c

if IsOnCircle(M3, c) = true then print('M3 ∈ c') end if;
                                     M3 ∈ c

if IsOnCircle(N1, c) = true then print('N1 ∈ c') end if;
                                     N1 ∈ c


if IsOnCircle(N2, c) = true then print('N2 ∈ c') end if;
                                     N2 ∈ c

if IsOnCircle(N3, c) = true then print('N3 ∈ c') end if;
                                     N3 ∈ c

dsegment(AH1, A, H1) : dsegment(BH2, B, H2) : dsegment(CH3, C,
H3) : dsegment(AM1, A, M1) : dsegment(BM2, B, M2) :
dsegment(CM3, C, M3) : draw([T(color = blue), c, H, H1, H2,
H3, N1, N2, N3, M1, M2, M3, AH1(color = green), BH2(color
= green), CH3(color = green), AM1(color = brown), BM2(color
= brown), CM3(color = brown)], printtext = true, axes = none,
title = 'DUONG TRON EULER');

```

DUONG TRON EULER



2. Ví dụ 2

Định lý FEUERBACH:

Đường tròn Euler tiếp xúc với các đường tròn nội tiếp và bàng tiếp

Đoạn lệnh đầu:

Cho tam giác ABC, vẽ đường tròn nội tiếp tâm H, các đường tròn bàng tiếp tâm I1, I2, I3 và đường tròn Euler tâm X. Kiểm tra sự tiếp xúc của đường tròn Euler với các đường tròn trên. MAPLE khẳng định các điều nêu trong Định lý.

```
> restart : with(geometry) : triangle(T, [point(A, 0, 0), point(B, 2, 4),
point(C, 7, 0)]) :
```

```
incircle(inc, T, 'centername' = I1) :
```

```
excircle(obj, T, [ex1(I1), ex2(I2), ex3(I3)]) :
```


```
EulerCircle(ec, T, 'centername' = X) :
```

```
if AreTangent(ec, inc) = true
    then print( `Duong tron Euler tiep xuc voi duong tron noi tiep` )
    end if;
if AreTangent(ec, ex1) = true
    then print( `Duong tron Euler tiep xuc voi duong tron bang tiep goc
A` ) end if; ;
if AreTangent(ec, ex2) = true
    then print( `Duong tron Euler tiep xuc voi duong tron bang tiep goc
B` ) end if; ;
if AreTangent(ec, ex3) = true
    then print( `Duong tron Euler tiep xuc voi duong tron bang tiep goc
C` ) end if; ;
```

Duong tron Euler tiep xuc voi duong tron noi tiep
Duong tron Euler tiep xuc voi duong tron bang tiep goc A
Duong tron Euler tiep xuc voi duong tron bang tiep goc B
Duong tron Euler tiep xuc voi duong tron bang tiep goc C

Đoạn lệnh tiếp mô tả bằng Hình vẽ.

```
> draw([inc(color = blue), ec(color = red), ex1(color = black),
ex2(color = black), ex3(color = black), T(color = blue), line(a,
[B, C]), line(b, [A, C]), line(c, [A, B])], filled = false, printtext
= true, scaling = constrained, title = "Định lý Feuerbach")
```



Đường tròn Euler (màu đỏ) tiếp xúc với đường tròn nội tiếp (màu xanh) và các đường tròn bàng tiếp (màu đen).

II. Sử dụng MAPLE chứng minh định lý trong trường hợp tổng quát

1. Phần mềm MAPLE tính toán được trên số, đặc biệt trên các ký tự (symbolic). Vì vậy có thể sử dụng MAPLE để chứng minh một số bài toán Hình học.

Muốn vậy cần phải xét trong một hệ tọa độ.

Chẳng hạn: Cho tam giác ABC,... cần chứng minh một điều gì đó.

Ta chọn hệ tọa độ vuông góc Oxy và phải cho các tọa độ tổng quát của 3 đỉnh $A(x_1, y_1)$, $B(x_2, y_2)$, $C(x_3, y_3)$, kèm theo các điều kiện để A, B, C không cùng nằm trên một đường thẳng.

Dựa trên các tham số này ta tiến hành giải quyết bài toán đặt ra.

Xét một trường hợp để xem MAPLE đã làm gì.

Cho 2 điểm A, B. Kiểm tra xem một điểm thứ 3 có nằm trên đường thẳng qua A, B không?

Gõ các lệnh sau:

> restart :

with(geometry) :

assume($a \neq b$) :

assume($c \neq d$) :

point(A, a, c) :

point(B, b, d) :

> line(l, [A, B]) :

Ta giả định các điểm được cho như sau: A(a, c), B(b, d) trong đó giả thiết $a \neq b$ và $c \neq d$.

Xây dựng đường thẳng (l) đi qua A và B. MAPLE thiết lập phương trình đường thẳng này theo công thức:

$$\frac{x-a}{b-a} = \frac{y-c}{d-c} \quad (*)$$

Khi cho điểm một điểm thứ 3, MAPLE sẽ thay tọa độ của điểm đó vào (*) và cho kết quả *true* hoặc *false*

Chẳng hạn:

> point($M, \frac{(a+b)}{2}, \frac{(c+d)}{2}$) :

IsOnLine(M, l);

true

Trường hợp này: M nằm trên (l)

Cho điểm P(a,d)

> point(P, a, d);

P

IsOnLine(P, l);

false

Trường hợp này P không nằm trên (l).

Cho điểm N

> point($N, \frac{(a+b)}{3}, \frac{(c+d)}{3}$):

IsOnLine(N, l);

IsOnLine: hint: unable to determine if $(-d+c)*(1/3*a+1/3*b)+(b-a)*(1/3*c+1/3*d)+d*a-c*b$ is zero

FAIL

MAPLE trả lời FAIL vì không xác định được biểu thức nêu trên có bằng 0 hay không.

Biểu thức trên là gì?

MAPLE lấy về phải trừ đi về trái của phương trình (*) và thay tọa độ của điểm N vào. Khi đó

$$\frac{\left(\frac{c}{3}+\frac{d}{3}\right)-c}{d-c} - \frac{\left(\frac{a}{3}+\frac{b}{3}\right)-a}{b-a} = \frac{\left(\frac{c}{3}+\frac{d}{3}\right)(b-a)-\left(\frac{a}{3}+\frac{b}{3}\right)(d-c)-c(b-a)+a(d-c)}{(b-a)(d-c)} = \frac{\left(\frac{a}{3}+\frac{b}{3}\right)(-d+c)+\left(\frac{c}{3}+\frac{d}{3}\right)(b-a)+ad-cb}{(b-a)(d-c)}$$

Biểu thức trên chính là tử số và MAPLE không biết biểu thức đó có triệt tiêu hay không.

Có quá nhiều điều kiện đặt ra cho các tham số cần xét đến.

Tương tự khi cho 3 điểm không thẳng hàng, có các tọa độ tổng quát. Ta vẽ được một đường tròn đi qua 3 điểm đó. MAPLE lập phương trình đường tròn theo các tham số trên.

Cho một điểm thứ 4, MAPLE thay tọa độ của điểm đó vào phương trình đường tròn để biết điểm này có nằm trên đường tròn hay không.....

2. Tuy nhiên với số tham số ít, ta có thể dùng MAPLE để chứng minh tổng quát được.

Sau đây, ta sử dụng MAPLE để chứng minh định lý sau:

Định lý:

Trong một tam giác ba điểm: trực tâm, trọng tâm và tâm đường tròn ngoại tiếp thẳng hàng.

Chứng minh:

Như cách thường nói “*Không làm giảm tổng quát, ta giả sử tam giác ABC có đỉnh A(a,b) với b>0, đỉnh B(0,0), đỉnh C(c,0) với c>0*”.

Kết quả này có được nhờ thực hiện các phép biến đổi: tịnh tiến, quay và lấy đối xứng.

Tam giác được xét có đỉnh B là gốc hệ tọa độ, cạnh BC nằm trên Ox và đỉnh A nằm phía trên trục hoành. Việc này làm giảm bớt các tham số khi xét.

Gõ đoạn lệnh sau: (Đó là toàn bộ bài Chứng minh)

restart : with(geometry) :

assume(b > 0) : assume(c > 0) :

triangle(T, [point(A, a, b), point(B, 0, 0), point(C, c, 0)]) :

orthocenter(H, T) :

centroid(G, T) :

circumcircle(c, T, 'centername'= O1) :

AreCollinear(G, H, O1);

Tam giác ABC có trực tâm H, tâm đường tròn ngoại tiếp O1, trọng tâm G.

MAPLE xác định các tọa độ của H, O1, G theo các tham số a, b, c.

Nhấn Enter liên tiếp.

Lệnh cuối MAPLE khẳng định G, H, O1 thẳng hàng.

`AreCollinear(G, H, O1);`

true

Định lý được chứng minh.

Muốn có một hình vẽ minh họa, cần cho a, b, c các giá trị cụ thể.

Thực ra 4 điểm: tâm đường tròn Euler, G, H, O1 cùng nằm trên một đường thẳng.

Gõ đoạn lệnh sau (Cho a = 2, b = 4, c = 7).

> `restart : with(geometry) :`

`triangle(T, [point(A, 2, 4), point(B, 0, 0), point(C, 7, 0)]) :`

`orthocenter(H, T) :`

`centroid(G, T) :`

`circumcircle(c, T, 'centername' = O1) :`

`EulerCircle(ec, T, 'centername' = Euler) :`

if `AreCollinear(O1, H, G) = true`

then `print('Ba điểm: trực tâm H, trọng tâm G và tâm đường tròn ngoại tiếp O1 thẳng hàng')` **end if;**

*Ba điểm: trực tâm H, trọng tâm G và tâm đường tròn ngoại tiếp O1
thẳng hàng*

Đoạn lệnh tiếp sau kiểm tra tâm đường tròn Euler nằm trên O1H, sau đó cho Hình vẽ minh họa

`line(l, [O1, H]) : IsOnLine(Euler, l);`

true

`draw([T(color = blue), c(thickness = 2, color = green), ec, O1,
l(color = brown), Euler, H, G], printtext = true, title
= 'Trục tâm H, trọng tâm G, tâm đường tròn ngoại tiếp O1 và tâm đường
tròn Euler thẳng hàng')`

